Appendix S1

The Construction and Application of a Population Physiologically Based Pharmacokinetic Model for Methadone in Beagles and Greyhounds

Trevor Elwell-Cuddy, Miao Li, Butch KuKanich, Zhoumeng Lin*

Institute of Computational Comparative Medicine (ICCM), Department of Anatomy and Physiology, College of Veterinary Medicine, Kansas State University, Manhattan, KS 66506, USA

Running title: PBPK model for methadone in dogs

Trevor Elwell-Cuddy: ElwellCudt@ksu.edu; Miao Li: miaoli@ksu.edu

Butch KuKanich: kukanich@ksu.edu; Zhoumeng Lin: zhoumeng@ksu.edu

* Corresponding author: Institute of Computational Comparative Medicine (ICCM), Department of Anatomy and Physiology, College of Veterinary Medicine, Kansas State University, 1800 Denison Avenue, P200 Mosier Hall, Manhattan, KS 66506, USA. E-mail: zhoumeng@ksu.edu; phone: +1-785-532-4087; fax: +1-785-532-4953.

Individual Beagle Model

```
METHOD RK4

STARTTIME = 0
STOPTIME=10
DT = 0.005
DTOUT = 0.1

; Physiological Parameters
; Blood flow rates
QCC = 12.9 ; cardiac output (L/h/kg) (Brown et al., 1997, pg. 441)
QLC = 0.046 ; Fraction of blood flow via hepatic artery to the liver (Brown et al., 1997, Table 26)
QKC = 0.173 ; Fraction of blood flow to the kidneys (Brown et al., 1997, Table 26)
QMC = 0.217 ; Fraction of blood flow to the muscle (Brown et al., 1997, Table 26)
QBC = 0.020 ; Fraction of blood flow to the brain (Brown et al., 1997, Table 26)
QLuC = 1 ; Fraction of blood flow to the lungs (Brown et al., 1997, Table 26)
QHC = 0.046 ; Fraction of blood flow to the heart (Brown et al., 1997, Table 26)
QRC = 1-QLC-QKC-QMC-QBC-QHC-QGC; Fraction of blood flow to the rest of body
QGC = 0.1 ; Fraction of blood flow to the GI tract (Delaney 1965, Table 3)

; Tissue volumes
BW = 17 ; Body weight (kg) (Ingvast-Larsson et al. 2010 17.0 kg for 0.4 mg/kg IV calibration and 0.4
mg/kg SC evaluation, KuKanich et al. 2005 10.15kg for 1.0 mg/kg IV evaluation)
VLC = 0.0329 ; Fractional liver tissue (Brown et al., 1997, Table 6)
VKC = 0.0055 ; Fractional kidney tissue (Brown et al., 1997, Table 6)
VMC = 0.4565 ; Fractional muscle tissue (Brown et al., 1997, Table 6)
VBC = 0.0078 ; Fractional brain tissue (Brown et al., 1997, Table 6)
VLuC = 0.0082 ; Fractional lung tissue (Brown et al., 1997, Table 6)
VHC = 0.0078 ; Fractional heart tissue (Brown et al., 1997, Table 6)
VGC = 0.0368 ; Fractional GI tract tissue (Brown et al., 1997, Table 6)
VbloodC = 0.082 ; Fractional blood (Brown et al., 1997, Table 21)
VartC = 0.2; Arterial blood volume, fraction of blood volume
VvenC = 1-VartC; Venous blood volume fraction of blood volume
VRC = 1-VLC-VKC-VMC-VBC-VLuC-VHC-VGC-VbloodC ; Fractional rest of body tissue (Brown et al.,
1997, Table 6)

; Mass Transfer Parameters (Chemical-specific parameters)
; Partition coefficients racemic methadone (PC tissue:plasma)
PM = 3.852 ; Muscle:plasma PC (Yang et al., 2006, Table II)
PLu = 42.46 ; Lung:plasma PC (Yang et al., 2006, Table II)
PBr = 2.076 ; Brain:plasma PC (Yang et al., 2006, Table II)
PH = 9.233 ; Heart:plasma PC (Yang et al., 2006, Table II)
PL = 19.46 ; Liver:plasma PC (Yang et al., 2006, Table II)
PG = 7.922 ; GITract:plasma PC (Yang et al., 2006, Table II)
PK = 10.61 ; Kidney:plasma PC (Yang et al., 2006, Table II)
PR = 5.44 ; restofbody:plasma PC (Average of other partition coefficients)

; Kinetic constants
; Oral absorption rate constants
Kst = 0 ; 1/h, gastric emptying rate constant
Ka = 0; 1/h, intestinal absorption rate constant
Kint = 0; 1/h, intestinal transit rate constant

; SC absorption rate constants
Ksc = 0.14 ; (1/h)
```

```
; IV injection time
Timeiv = 0.01; IV injection time (h) based on Lin et al. 2014 & Leavens et al. 2012

; Percentage Plasma Protein Binding unitless
PB = 0.648; Percentage of drug bound to plasma proteins; based on Derendorf & Garrett, 1983

; Eimination rate constants
KurineC = 0.8 ; L/h/kg
KmC = 0.02 ; /(h*kg)

; Parameters for various exposure scenarios
PDOSEiv = 0.4 ; (mg/kg)
PDOSEsc = 0; (mg/kg)
PDOSEoral = 0 ; (mg/kg)

; Cardiac output and blood flows to tissues (L/h)
QC = QCC*BW ; cardiac output
QL = QLC*QC ; liver
QK = QKC*QC ; kidneys
QB = QBC*QC ; brain
QM = QMC*QC ; muscle
QR = QRC*QC ; rest of body
QG = QGC*QC ; GI Tract
QH = QHC*QC ; heart

; Tissue volumes (L)
VL = VLC*BW ; Liver
VK = VKC*BW ; Kidneys
VM = VMC*BW ; Muscle
VLu = VLuC*BW ; Lungs
VB = VBC*BW ; Brain
VH = VHC*BW ; Heart
VG = VGC*BW ; GI Tract
VR = VRC*BW ; Rest of body
Vblood = VbloodC*BW ; Blood
Vven = VvenC*Vblood ; Venous Blood
Vart = VartC*Vblood ; Arterial Blood

; Dosing
DOSEoral = PDOSEoral*BW ; (mg)
DOSEiv = PDOSEiv*BW ; (mg)
DOSEsc = PDOSEsc*BW ; (mg)

; Dosing, oral gavage
tlen = 0.1 ; length of oral gavage exposure (h)


RAST = -Kst*AST; rate of change of amount in stomach (mg/h)
d/dt(AST) = RAST; derivative of amount in stomach
init AST = DOSEoral; initial amount in stomach (mg)
RAI = Kst*AST-Ka*AI-Kint*AI; rate of change of amount of drug in the intestine (mg/h)
Rcolon = Kint*AI; rate of change of amount in colon (mg/h)
d/dt(Acolon) = Rcolon; derivative of amount in colon
init Acolon = 0; initial amount in colon (mg)
d/dt(AI) = RAI; derivative of amount in intestine
init AI = 0; initial amount in intestine (mg)
```

RAO = Ka*AI; intestinal absorption rate (mg/h)
d/dt(AAO) = RAO; derivative of the amount absorbed via oral exposure (mg)
init AAO = 0; initial amount absorbed via oral exposure (mg)

; Dosing, SC, subcutaneous
Rsc = Ksc*Amtsitesc; (mg/h); Absorption rate (mg/h)
Rsitesc = -Rsc; (mg/h); rate of change in the amount of absorbable methadone in the injection site (mg/h)
d/dt(Amtsitesc) = Rsitesc; (mg); derivative of the amount of absorbable methadone that remains in the injection site
init Amtsitesc = DOSEsc; (mg); initial amount of absorbable methadone at the injection site
d/dt (Absorbsc) = Rsc; (mg); derivative of the amount of methadone absorbed
init Absorbsc = 0; initial amount of methadone absorbed

; methadone iv injection to the venous
IVR = DOSEiv/Timeiv; injection dose/IV injection time, mg/h
Riv = IVR*(1.-step(1,Timeiv)); injection rate (mg/h)
d/dt(Aiv) = Riv; derivative of administered amount (mg)
init Aiv = 0; initial administered amount (mg)

; Eimination rate constants
Kurine = KurineC*BW ; L/h
Kmetabolites = KmC*BW ; /h

; methadone in blood compartment, flow-limited model
; venous blood
RV = (QL*CVL+QK*CVK+QM*CVM+QH*CVH+QB*CVB+QR*CVR+Riv+Rsc)-QC*CV; rate of change of methadone in venous blood (mg/h)
d/dt(AV) = RV; amount in the venous blood (mg)
init AV = 0; initial amount in the venous blood (mg)
CV=AV/Vven; concentration in the venous blood (mg/L)
CVppb=CV*1000; conversion from ppm to ppb
CVfree = CV*(1-PB); CVfree concentration of unbound drug in the venous blood (mg/L)
d/dt(AUCCV) = CV; derivative of the area under the curve of methadone concentration in the venous blood
init AUCCV = 0; initial area under the curve concentration of methadone in the venous blood (mg/mL)*h
AUCCVPPB = AUCCV*1000; conversion from ppm to ppb

RA = QC*CVLu-QC*CAfree ; rate of change in arterial blood (mg/h)
d/dt(AA) = RA; derivative of amount in arterial blood (mg)
init AA = 0; initial amount of methadone in arterial blood (mg)
CA = AA/Vart; concentration in the arterial blood (mg/L)
CAfree = CA*(1-PB); amount of unbound methadone in the arterial blood (mg)

; methadone in muscle compartment, flow-limited model
RM = QM*(CAfree-CVM); rate of change of methadone in the muscle compartment (mg/h)
d/dt(AM) = RM; deriviative of the amount of methadone in the muscle compartment (mg)
init AM = 0; initial amount of methadone in the muscle compartment (mg)
CM = AM/VM; concentration of methadone in the muscle compartment (mg)
CVM = AM/(VM*PM); amount of methadone in the blood of the muscle compartment (mg)
d/dt(AUCCM) = CM; derivative of the area under the curve of methadone concentration in the muscle (mg/mL)*h
init AUCCM = 0; initial area under the curve concentration of methadone (mg/mL)*h
AUCCMPPB = AUCCM*1000; conversion from ppm to ppb

; methadone in lung compartment, flow-limited model
RLu = QC*(CV-CVLu); rate of change of methadone in the lung compartment (mg/h)

d/dt(ALu) = RLu; deriviative of the amount of methadone in the lung compartment (mg)
init ALu = 0; initial amount of methadone in the lung compartment (mg)
CLu = ALu/VLu; concentration of methadone in the lung compartment (mg)
CVLu = ALu/(VLu*PLu); amount of methadone in the blood of the lung compartment (mg)
d/dt(AUCCLu) = CLu; derivative of the area under the curve of methadone concentration in the lung (mg/mL)*h
init AUCCLu = 0; initial area under the curve concentration of methadone in the lung (mg/mL)*h

; methadone in rest of body compartment, flow-limited model
RR = QR*(CAfree-CVR); rate of change of methadone in the rest of body compartment (mg/h)
d/dt(AR) = RR; deriviative of the amount of methadone in the rest of body compartment (mg)
init AR = 0; initial amount of methadone in the rest of body compartment (mg)
CR = AR/VR; concentration of methadone in the rest of body compartment (mg)
CVR = AR/(VR*PR); amount of methadone in the blood of the rest of body compartment (mg)
d/dt(AUCCR)= CR; derivative of the area under the curve of methadone concentration in the rest of body (mg/mL)*h
init AUCCR = 0; initial area under the curve concentration of methadone in the rest of body (mg/mL)*h

; methadone in brain compartment, flow-limited model
RB = QB*(CAfree-CVB); rate of change of methadone in the brain compartment (mg/h)
d/dt(AB) = RB; deriviative of the amount of methadone in the brain compartment (mg)
init AB = 0; initial amount of methadone in the brain compartment (mg)
CB = AB/VB; concentration of methadone in the brain compartment (mg)
CVB = AB/(VB*PBr); amount of methadone in the blood of the brain compartment (mg)
d/dt(AUCCB) = CB; derivative of the area under the curve of methadone concentration in the brain (mg/mL)*h
init AUCCB = 0; initial area under the curve concentration of methadone in the brain (mg/mL)*h

; methadone in heart compartment, flow-limited model
RH = QH*(CAfree-CVH); rate of change of methadone in the heart compartment (mg/h)
d/dt(AH) = RH; deriviative of the amount of methadone in the heart compartment (mg)
init AH = 0; initial amount of methadone in the heart compartment (mg)
CH = AH/VH; concentration of methadone in the heart compartment (mg)
CVH = AH/(VH*PH); amount of methadone in the blood of the heart compartment (mg)
d/dt(AUCCH) = CH; derivative of the area under the curve of methadone concentration in the heart (mg/mL)*h
init AUCCH = 0; initial area under the curve concentration of methadone in the heart (mg/mL)*h

; methadone in liver compartment, flow-limited model
RL = QL*(CAfree-CVL)+QG*CVG+RAO-Rmetabolites; rate of change of methadone in the liver compartment (mg/h)
d/dt(AL) = RL; deriviative of the amount of methadone in the liver compartment (mg)
init AL = 0; initial amount of methadone in the liver compartment (mg)
CL = AL/VL; concentration of methadone in the liver compartment (mg)
CVL = AL/(VL*PL); amount of methadone in the blood of the liver compartment (mg)
d/dt(AUCCL) = CL; derivative of the area under the curve of methadone concentration in the liver (mg/mL)*h
init AUCCL = 0; initial area under the curve concentration of methadone in the liver (mg/mL)*h

; metabolic excretion of methadone
Rmetabolites = Kmetabolites*CL*VL; rate of change of amount of metabolized methadone
d/dt(Ametabolites) = Rmetabolites; derivative of the amount of metabolized methadone
init Ametabolites = 0; initial amount of metabolized methadone

; methadone in GI Tract compartment, flow-limited model
RG = QG*(CAfree-CVG); rate of change of methadone in the GI tract compartment (mg/h)

d/dt(AG) = RG; deriviative of the amount of methadone in the GI tract compartment (mg)
init AG = 0; initial amount of methadone in the GI tract compartment (mg)
CG = AG/VG; concentration of methadone in the GI tract compartment (mg)
CVG = AG/(VG*PG); amount of methadone in the blood of the GI tract compartment (mg)
d/dt(AUCCG) = CG; derivative of the area under the curve of methadone concentration in the GI tract (mg/mL)*h
init AUCCG = 0; initial area under the curve concentration of methadone in the GI tract (mg/mL)*h

; methadone in kidney compartment, flow-limited model
RK = QK*(CAfree-CVK)-Rurine; rate of change of methadone in the kidney compartment (mg/h)
d/dt (AK) = RK; deriviative of the amount of methadone in the kidney compartment (mg)
init AK = 0; initial amount of methadone in the kidney compartment (mg)
CK = AK/VK; concentration of methadone in the kidney compartment (mg)
CVK = AK/(VK*PK); amount of methadone in the blood of the kidney compartment (mg)
d/dt(AUCCK) = CK; derivative of the area under the curve of methadone concentration in the kidney (mg/mL)*h
init AUCCK = 0; initial area under the curve concentration of methadone in the kidney (mg/mL)*h

; urinary excretion of methadone
Rurine = Kurine*CVK; rate of change of amount of methadone in the urine
d/dt(Aurine) = Rurine; derivative of the amount of methadone in the urine
init Aurine = 0; initial amount of methadone in the urine

; Mass balance
Qbal = QC-QL-QK-QM-QB-QH-QR-QG; cardiac output balance
Tmass = AA+AV+AM+ALu+AB+AH+AL+AG+AK+AR+Aurine+Ametabolites; total methadone balance
Bal =AAO+Aiv+Absorbsc-Tmass; mass balance

Population Beagle Model

{Monte Carlo analysis based on methadone PBPK model for Beagles and Greyhounds (flow-limited model, linear metabolism equation, plasma protein binding). The PBPK model code is based on the Miao Li Penicillin PBPK model for cattle and the oxytetracycline model from Zhoumeng Lin}

METHOD Stiff

STARTTIME = 0
STOPTIME= 100; h,24
DT = 0.00025
DTOUT = 0.1

; Physiological Parameters
; Blood flow rates
QCC = 12.9 ; cardiac output (L/h/kg) (Brown et al., 1997, pg. 441)

; Fraction of blood flow to organs (unitless)
QLC = 0.046 ; Fraction of blood flow via hepatic artery to the liver (Brown et al., 1997, Table 26)
init QKC = 0.173 ; Fraction of blood flow to the kidneys (Brown et al., 1997, Table 26)
init QMC = 0.217 ; Fraction of blood flow to the muscle (Brown et al., 1997, Table 26)
init QBC = 0.020 ; Fraction of blood flow to the brain (Brown et al., 1997, Table 26)
QLuC = 1 ; Fraction of blood flow to the lungs (Brown et al., 1997, Table 26)
init QHC = 0.046 ; Fraction of blood flow to the heart (Brown et al., 1997, Table 26)
init QRC = 1-QLC-QKC-QMC-QBC-QHC-QGC; Fraction of blood flow to the rest of body
QGC = 0.1 ; Fraction of blood flow to the GI tract (Delaney 1965, Table 3)

; Tissue volumes
BW = 17 ; Body weight (kg) (Ingvast-Larsson et al. 2010 17.0 kg for 0.4 mg/kg IV calibration and 0.4 mg/kg SC evaluation, KuKanich et al. 2005 10.15kg for 1.0 mg/kg IV evaluation)

; Fractional organ tissue volumes (unitless)
VLC = 0.0329 ; Fractional liver tissue (Brown et al., 1997, Table 6)
init VKC = 0.0055 ; Fractional kidney tissue (Brown et al., 1997, Table 6)
init VMC = 0.4565 ; Fractional muscle tissue (Brown et al., 1997, Table 6)
init VBC = 0.0078 ; Fractional brain tissue (Brown et al., 1997, Table 6)
init VLuC = 0.0082 ; Fractional lung tissue (Brown et al., 1997, Table 6)
init VHC = 0.0078 ; Fractional heart tissue (Brown et al., 1997, Table 6)
init VGC = 0.0368 ; Fractional GI tract tissue (Brown et al., 1997, Table 6)
init VbloodC = 0.082 ; Fractional blood (Brown et al., 1997, Table 21)
VartC = 0.2; Arterial blood volume, fraction of blood volume
VvenC = 1-VartC; Venous blood volume fraction of blood volume
init VRC = 1-VLC-VKC-VMC-VBC-VLuC-VHC-VGC-VbloodC ; Fractional rest of body tissue (Brown et al., 1997, Table 6)

; Mass Transfer Parameters (Chemical-specific parameters)
; Partition coefficients racemic methadone (PC tissue:plasma)
PM = 3.852 ; Muscle:plasma PC (Yang et al., 2006, Table II)
PLu = 42.46 ; Lung:plasma PC (Yang et al., 2006, Table II)
PBr = 2.076 ; Brain:plasma PC (Yang et al., 2006, Table II)
PH = 9.233 ; Heart:plasma PC (Yang et al., 2006, Table II)
PL = 19.46 ; Liver:plasma PC (Yang et al., 2006, Table II)
PG = 7.922 ; GITract:plasma PC (Yang et al., 2006, Table II)
PK = 10.61 ; Kidney:plasma PC (Yang et al., 2006, Table II)
PR = 5.44 ; restofbody:plasma PC (Average of other partition coefficients)

; Kinetic constants
; SC absorption rate constants
Ksc = 0.14 ; (1/h)

; IV injection time
Timeiv = 0.01; IV injection time (h) based on Lin et al. 2014 & Leavens et al. 2012

; Percentage Plasma Protein Binding unitless
PB = 0.648; Percentage of drug bound to plasma proteins; based on Derendorf & Garrett, 1983
Free = 1-PB; Percentage of drug not bound to plasma protein

; Elimination rate constants
KurineC = 0.8 ; L/h/kg; urinary elimination rate constant
KmC = 0.02 ; /(h*kg); metabolic rate constant

; Parameters for various exposure scenarios
PDOSEiv = 0.4; (mg/kg)
PDOSEsc = 0; (mg/kg)

{Standard Deviation of Parameters}
BW_sd = 3.100; Standard Deviation of BW
VLC_sd = 0.002; Standard Deviation of VLC
QLC_sd = 0.089; Standard Deviation of QLC
QGC_sd = 0.030; Standard Deviation of QGC
PBr_sd = 0.415; Standard Deviation of PBr
PL_sd = 3.892; Standard Deviation of PL
KmC_sd = 0.006; Standard Deviation of KmC
KurineC_sd = 0.240; Standard Deviation of KurineC

{Generation of Parameters based on Normal Distribution}
init BWm = Normal(BW, BW_sd); Generation of the BWm based on normal distribution
init VLCm = Normal(VLC, VLC_sd); Generation of the VLCm based on normal distribution
init QLCm = Normal(QLC, QLC_sd); Generation of the QLCm based on normal distribution
init QGCm = Normal(QGC, QGC_sd); Generation of the QGCm based on normal distribution

; Assignment of the Values to Parameters
next BWm=BWm; assignment of first created value to BWm, without this step BWm will change at each integration time step

; Creation of Adjust Factor
AdjustF = QLCm+QKC+QMC+QBC+QHC+QGCm+QRC; Adjust factor to keep the sum of blood flow fractions to 1
AdjustF1 = VLCm+VKC+VMC+VBC+VLuC+VHC+VGC+VbloodC+VRC; Adjust factor to keep sum of organ tissue volumes at 1

; Creation of Adjusted Parameters
next VLCm = VLCm/AdjustF1; Adjustment of VLCm based on the adjust factor
next VKC = VKC/AdjustF1; Adjustment of VKC based on the adjust factor
next VMC = VMC/AdjustF1; Adjustment of VMC based on the adjust factor
next VBC = VBC/AdjustF1; Adjustment of VBC based on the adjust factor
next VLuC = VLuC/AdjustF1; Adjustment of VLuC based on the adjust factor
next VHC = VHC/AdjustF1; Adjustment of VHC based on the adjust factor
next VGC = VGC/AdjustF1; Adjustment of VGC based on the adjust factor
next VbloodC = VbloodC/AdjustF1; Adjustment of VbloodC based on the adjust factor
next VRC = VRC/AdjustF1; Adjustment of VRC based on the adjust factor
next QLCm = QLCm/AdjustF; Adjustment of QLCm based on the adjust factor

```
next QGCm = QGCm/AdjustF; Adjustment of QGCm based on the adjust factor
next QKC = QKC/AdjustF; Adjustment of QKC based on the adjust factor
next QMC = QMC/AdjustF; Adjustment of QMC based on the adjust factor
next QBC = QBC/AdjustF; Adjustment of QBC based on the adjust factor
next QHC = QHC/AdjustF; Adjustment of QHC based on the adjust factor
next QRC = QRC/AdjustF; Adjustment of QRC based on the adjust factor

{Lognormal Transformation of Parameters}
PBr_ln = logn(PBr^2/(PBr_sd^2+PBr^2)^0.5); lognormal transformation of PBr values
PBr_lnsd = logn(1+PBr_sd^2/PBr^2); lognormal transformation of PBr standard deviation
PL_ln = logn(PL^2/(PL_sd^2+PL^2)^0.5); lognormal transformation of PL values
PL_lnsd = logn(1+PL_sd^2/PL^2); lognormal transformation of PL standard deviation
KmC_ln = logn(KmC^2/(KmC_sd^2+KmC^2)^0.5); lognormal transformation of KmC values
KmC_lnsd = logn(1+KmC_sd^2/KmC^2); lognormal transformation of KmC standard deviation
KurineC_ln = logn(KurineC^2/(KurineC_sd^2+KurineC^2)^0.5); lognormal transformation of KurineC
values
KurineC_lnsd = logn(1+KurineC_sd^2/KurineC^2); lognormal transformation of KurineC standard
deviation

{Creation of Parameters based on Lognormal Distribution}
init PBrm = exp(Normal(PBr_ln, PBr_lnsd)) next PBrm = PBrm; Generation of PBrm based on lognormal
distribution
init PLm = exp(Normal(PL_ln, PL_lnsd)) next PLm = PLm; Generation of PLm based on lognormal
distribution
init KmCm = exp(Normal(KmC_ln, KmC_lnsd)) next KmCm = KmCm; Generation of KmCm based on
lognormal distribution
init KurineCm = exp(Normal(KurineC_ln, KurineC_lnsd)) next KurineCm = KurineCm; Generation of
KurineCm based on lognormal distribution

{limit the parameter values within the lower and upper bounds}
limit BWm >= 10.924; lower bound of BWm
limit BWm <= 23.076; upper bound of BWm
limit VLCm >= 0.028; lower bound of VLCm
limit VLCm <= 0.038; upper bound of VLCm
limit VKC >= 0.004; lower bound of VKC
limit VKC <= 0.007; upper bound of VKC
limit VMC >= 0.348; lower bound of VMC
limit VMC <= 0.565; upper bound of VMC
limit VBC >= 0.005; lower bound of VBC
limit VBC <= 0.011; upper bound of VBC
limit VLuC >= 0.007; lower bound of VLuC
limit VLuC <= 0.009; upper bound of VLuC
limit VHC >= 0.023; lower bound of VHC
limit VHC <= 0.050; upper bound of VHC
limit VbloodC >= 0.034; lower bound of VbloodC
limit VbloodC <= 0.130; upper bound of VbloodC
limit QLCm >= 0.122; lower bound of QLCm
limit QLCm <= 0.472; upper bound of QLCm
limit QKC >= 0.088; lower bound of QKC
limit QKC <= 0.258; upper bound of QKC
limit QMC >= 0.021; lower bound of QMC
limit QMC <= 0.413; upper bound of QMC
limit QBC >= 0.013; lower bound of QBC
limit QBC <= 0.027; upper bound of QBC
limit QHC >= 0.015; lower bound of QHC
limit QHC <= 0.077; upper bound of QHC
```

```
limit QGC >= 0.041; lower bound of QGC
limit QGC <= 0.159; upper bound of QGC
limit PBr >= 1.381; lower bound of PBr
limit PBr <= 3.001; upper bound of PBr
limit PL >= 12.94; lower bound of PL
limit PL <= 28.13; upper bound of PL
limit KmC >= 0.011; lower bound of KmC
limit KmC <= 0.034; upper bound of KmC
limit KurineC >= 0.431; lower bound of KurineC
limit KurineC <= 1.362; upper bound of KurineC

; Cardiac output and blood flows to tissues (L/h)
QC = QCC*BWm ; cardiac output
QL = QLCm*QC ; liver
QK = QKC*QC ; kidneys
QB = QBC*QC ; brain
QM = QMC*QC ; muscle
QR = QRC*QC ; rest of body
QG = QGCm*QC ; GI Tract
QH = QHC*QC ; heart

; Tissue volumes (L)
VL = VLCm*BWm ; Liver
VK = VKC*BWm ; Kidneys
VM = VMC*BWm ; Muscle
VLu = VLuC*BWm ; Lungs
VB = VBC*BWm ; Brain
VH = VHC*BWm ; Heart
VG = VGC*BWm ; GI Tract
VR = VRC*BWm ; Rest of body
Vblood = VbloodC*BWm ; Blood
Vven = VvenC*VbloodC ; Venous Blood
Vart = VartC*VbloodC ; Arterial Blood

; Dosing
DOSEiv = PDOSEiv*BWm ; (mg)
DOSEsc = PDOSEsc*BWm ; (mg)

; Dosing, SC, subcutaneous
Rsc = Ksc*Amtsitesc; (mg/h)
Rsitesc = -Rsc; (mg/h)
d/dt(Amtsitesc) = Rsitesc; (mg)
init Amtsitesc = DOSEsc; (mg)
d/dt (Absorbsc) = Rsc; (mg)
init Absorbsc = 0;  initial amount of methadone absorbed

; methadone iv injection to the venous
IVR = Doseiv/Timeiv; injection dose/IV injection time, mg/h
Riv = IVR*(1.-step(1,Timeiv)); injection rate (mg/h)
d/dt(Aiv) = Riv; derivative of administered amount (mg)
init Aiv = 0; initial administered amount (mg)

; Eimination rate constants
Kurine = KurineCm*BWm ; L/h
Kmetabolites = KmCm*BWm ; /h
```

; methadone in blood compartment, flow-limited model
; venous blood
RV = (QL*CVL+QK*CVK+QM*CVM+QH*CVH+QB*CVB+QR*CVR+Riv+Rsc)-QC*CV; the changing rate in the venous blood
d/dt(AV) = RV; amount in the venous blood (mg)
init AV = 0; initial amount in the venous blood (mg)
CV=AV/Vven; concentration in the venous blood (mg/L)
CVppb=CV*1000; conversion from ppm to ppb
CVfree = CV*(1-PB); CVfree concentration of unbound drug in the venous blood (mg/L)
d/dt(AUCCV) = CV; derivative of the area under the curve of methadone concentration in the venous blood
init AUCCV = 0; initial area under the curve concentration of methadone in the venous blood (mg/mL)*h
AUCCVPPB = AUCCV*1000; conversion from ppm to ppb

RA = QC*CVLu-QC*CAfree ; rate of change in arterial blood (mg/h)
d/dt(AA) = RA; derivative of amount in arterial blood (mg)
init AA = 0; initial amount of methadone in arterial blood (mg)
CA = AA/Vart; concentration in the arterial blood (mg/L)
CAfree = CA*(1-PB); amount of unbound methadone in the arterial blood (mg)

; methadone in muscle compartment, flow-limited model
RM = QM*(CAfree-CVM); rate of change of methadone in the muscle compartment (mg/h)
d/dt(AM) = RM; deriviative of the amount of methadone in the muscle compartment (mg)
init AM = 0; initial amount of methadone in the muscle compartment (mg)
CM = AM/VM; concentration of methadone in the muscle compartment (mg)
CVM = AM/(VM*PM); amount of methadone in the blood of the muscle compartment (mg)
d/dt(AUCCM) = CM; derivative of the area under the curve of methadone concentration in the muscle (mg/mL)*h
init AUCCM = 0; initial area under the curve concentration of methadone (mg/mL)*h
AUCCMPPB = AUCCM*1000; conversion from ppm to ppb

; methadone in lung compartment, flow-limited model
RLu = QC*(CV-CVLu); rate of change of methadone in the lung compartment (mg/h)
d/dt(ALu) = RLu; deriviative of the amount of methadone in the lung compartment (mg)
init ALu = 0; initial amount of methadone in the lung compartment (mg)
CLu = ALu/VLu; concentration of methadone in the lung compartment (mg)
CVLu = ALu/(VLu*PLu); amount of methadone in the blood of the lung compartment (mg)
d/dt(AUCCLu) = CLu; derivative of the area under the curve of methadone concentration in the lung (mg/mL)*h
init AUCCLu = 0; initial area under the curve concentration of methadone in the lung (mg/mL)*h

; methadone in rest of body compartment, flow-limited model
RR = QR*(CAfree-CVR); rate of change of methadone in the rest of body compartment (mg/h)
d/dt(AR) = RR; deriviative of the amount of methadone in the rest of body compartment (mg)
init AR = 0; initial amount of methadone in the rest of body compartment (mg)
CR = AR/VR; concentration of methadone in the rest of body compartment (mg)
CVR = AR/(VR*PR); amount of methadone in the blood of the rest of body compartment (mg)
d/dt(AUCCR)= CR; derivative of the area under the curve of methadone concentration in the rest of body (mg/mL)*h
init AUCCR = 0; initial area under the curve concentration of methadone in the rest of body (mg/mL)*h

; methadone in brain compartment, flow-limited model
RB = QB*(CAfree-CVB); rate of change of methadone in the brain compartment (mg/h)
d/dt(AB) = RB; deriviative of the amount of methadone in the brain compartment (mg)
init AB = 0; initial amount of methadone in the brain compartment (mg)
CB = AB/VB; concentration of methadone in the brain compartment (mg)

```
CBppb=CB*1000; conversion from ppm to ppb
CVB = AB/(VB*PBrm); amount of methadone in the blood of the brain compartment (mg)
d/dt(AUCCB) = CB; derivative of the area under the curve of methadone concentration in the brain (mg/mL)*h
init AUCCB = 0; initial area under the curve concentration of methadone in the brain (mg/mL)*h

; methadone in heart compartment, flow-limited model
RH = QH*(CAfree-CVH); rate of change of methadone in the heart compartment (mg/h)
d/dt(AH) = RH; deriviative of the amount of methadone in the heart compartment (mg)
init AH = 0;  initial amount of methadone in the heart compartment (mg)
CH = AH/VH; concentration of methadone in the heart compartment (mg)
CVH = AH/(VH*PH); amount of methadone in the blood of the heart compartment (mg)
d/dt(AUCCH) = CH; derivative of the area under the curve of methadone concentration in the heart (mg/mL)*h
init AUCCH = 0; initial area under the curve concentration of methadone in the heart (mg/mL)*h

; methadone in liver compartment, flow-limited model
RL = QL*(CAfree-CVL)+QG*CVG-Rmetabolites; rate of change of methadone in the liver compartment (mg/h)
d/dt(AL) = RL; deriviative of the amount of methadone in the liver compartment (mg)
init AL = 0; initial amount of methadone in the liver compartment (mg)
CL = AL/VL; concentration of methadone in the liver compartment (mg)
CVL = AL/(VL*PLm); amount of methadone in the blood of the liver compartment (mg)
d/dt(AUCCL) = CL; derivative of the area under the curve of methadone concentration in the liver (mg/mL)*h
init AUCCL = 0;  initial area under the curve concentration of methadone in the liver (mg/mL)*h

; metabolic excretion of methadone
Rmetabolites = Kmetabolites*CL*VL; Kmetabolites*CL*VL; rate of change of amount of metabolized methadone
d/dt(Ametabolites) = Rmetabolites; Rmetabolites; derivative of the amount of metabolized methadone
init Ametabolites = 0; initial amount of metabolized methadone

; methadone in GI Tract compartment, flow-limited model
RG = QG*(CAfree-CVG); rate of change of methadone in the GI tract compartment (mg/h)
d/dt(AG) = RG; deriviative of the amount of methadone in the GI tract compartment (mg)
init AG = 0;  initial amount of methadone in the GI tract compartment (mg)
CG = AG/VG; concentration of methadone in the GI tract compartment (mg)
CVG = AG/(VG*PG); amount of methadone in the blood of the GI tract compartment (mg)
d/dt(AUCCG) = CG; derivative of the area under the curve of methadone concentration in the GI tract (mg/mL)*h
init AUCCG = 0; initial area under the curve concentration of methadone in the GI tract (mg/mL)*h

; methadone in kidney compartment, flow-limited model
RK = QK*(CAfree-CVK)-Rurine; rate of change of methadone in the kidney compartment (mg/h)
d/dt (AK) = RK; deriviative of the amount of methadone in the kidney compartment (mg)
init AK = 0;  initial amount of methadone in the kidney compartment (mg)
CK = AK/VK; concentration of methadone in the kidney compartment (mg)
CVK = AK/(VK*PK); amount of methadone in the blood of the kidney compartment (mg)
d/dt(AUCCK) = CK; derivative of the area under the curve of methadone concentration in the kidney (mg/mL)*h
init AUCCK = 0;  initial area under the curve concentration of methadone in the kidney (mg/mL)*h

; urinary excretion of methadone
Rurine = Kurine*CVK; rate of change of amount of methadone in the urine
d/dt(Aurine) = Rurine; derivative of the amount of methadone in the urine
```

```
init Aurine = 0;  initial amount of methadone in the urine

; Mass balance
Qbal = QC-QL-QK-QM-QB-QH-QR-QG; cardiac output balance
Tmass = AA+AV+AM+ALu+AB+AH+AL+AG+AK+AR+Aurine+Ametabolites; total methadone balance
Bal = Aiv+Absorbsc-Tmass; mass balance
```